

```

cd "D:\Jason\workshop\Regression 2019\final"

log using "D:\Jason\workshop\Regression 2019\final\regression1.log", replace

set more 1

use "D:\Jason\workshop\Regression 2019\final\regression.dta", clear

*****
* Continuous dependent variable
*****

*****
* Questions 1:
*****

*****
* 1.1 Regression without complex survey data
*****

*****
* 1.1.0 Regression without using the weight variable
*****

reg illness sex age

*****
* 1.1.1. Regression using a personal weight variable for the whole sample
*****

reg illness sex age [pw=finalwgt]

*****
* 1,1,2. Regression using a personal weight variable and adjusted for
*dependent observations for the whole sample
*****

reg illness sex age [pw=finalwgt], cluster(company_id)

*****
* 1.1.3. Regression using a personal weight variable and adjusted for
*dependent observations for white respondents
*****

reg illness sex age if race ==1 [pw=finalwgt], cluster(company_id)

*****
* 1.2. Regression with complex survey data
*****
svydes
svyset, clear
svyset psuid [pweight=finalwgt], strata(stratid)

*****
* 1.2.1. Regression using complex survey data for the whole sample
*****

svy: reg illness sex age

*****
* 1,2,2. Regression using complex survey data and adjusted for
*dependent observations for the whole sample
*****

*Cluster option is not allowed with the svy command

```

```

*****
* 1.2.3. Regression using a personal weight variable and adjusted for
*dependent observations for white respondents only
*****

*correct specification*

gen white = 0
replace white =1 if race ==1
svy, subpop(white): reg illness sex age

*incorrect specification*
svy: reg illness sex age if race ==1 /* incorrect specification */

```

```

*****
* Questions 2 and 3: the equality and linear combination of regression coefficients
*****

```

```

*****
* 2.1 Modifying Mdoel 1.1.3
*****
reg illness sex age if race ==1 [pw=finalwgt], cluster(company_id)

*testing the equlaity of two regression coefficients
test sex = age

*testing the hypothesis that variables cancel out each other
test sex = -age

```

```

*****
* 2.2 Regression Analysis Using Results from Model 1.2.3.
*****

svy, subpop(white): reg illness sex age hlthstat

*testing the equlaity of two or three regression coefficients
test sex = age

*testing the equlaity of two or three regression coefficients
test sex = -age

```

```

*****
* Questions 4: testing the interaction effects
*****

```

```

*****
* 4.1.1 A two-way interaction: Modify the model 1.1.2
*****
reg illness i.sex##c.age, cluster(company_id)
reg, coeflegend
test _b[2.sex#c.age] =0

margins i.sex, at(age=(20(10)80))
marginsplot, yline(0)

```

```

*****
* 4.1.2 A three-way interactdion: Modify the model 1.2.3.
*****
reg illness i.sex##c.age##i.race [pw=finalwgt], cluster(company_id)
reg, coeflegend
test _b[2.sex#c.age] =0

```

```
margins i.sex , over(i.race) at(age=(20(10)80))
marginsplot, yline(0)
```

```
*****
* 4.2.1 A two-way interaction Modify the model 1.2.1.
*****
svy: reg illness i.sex##c.age
reg, coeflegend
test _b[2.sex#c.age] =0
```

```
margins i.sex, at(age=(20(10)80)) vce(unconditional)
marginsplot, yline(0)
```

```
*****
* 4.2.1 A three-way interactdion: Modify the model 1.2.1
*****
```

```
svy: reg illness i.sex##c.age##i.race
reg, coeflegend
```

```
margins i.sex , over(i.race) at(age=(20(10)80)) vce(unconditional)

marginsplot, yline(0)
```

```
*****
* Questions 5: testing the total effects of X
*****
```

```
*****
* 5.1 Regression using a simple survey data and a three-way interaction term
*****
```

```
reg illness i.sex##c.age##hlthstat [pw=finalwgt], cluster(company_id)
margins i.sex, atmeans
margins, dydx(i.sex) atmeans
```

```
*****
* 5.1 Regression using complex survey data and a three-way interaction term
*****
```

```
svy: reg illness i.sex##c.age##hlthstat
margins i.sex, atmeans vce(unconditional)
margins, dydx(i.sex) atmeans vce(unconditional)
```

log close